NAG Library Routine Document

F07GEF (DPPTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07GEF (DPPTRS) solves a real symmetric positive definite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07GDF (DPPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07GEF (UPLO, N, NRHS, AP, B, LDB, INFO)
INTEGER             N, NRHS, LDB, INFO
REAL (KIND=nag_wp) AP(*), B(LDB,*)
CHARACTER(1)        UPLO
```

The routine may be called by its LAPACK name ***dpptrs***.

3 Description

F07GEF (DPPTRS) is used to solve a real symmetric positive definite system of linear equations $AX = B$, the routine must be preceded by a call to F07GDF (DPPTRF) which computes the Cholesky factorization of A, using packed storage. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^{\mathrm{T}}U$, where U is upper triangular; the solution X is computed by solving $U^{\mathrm{T}}Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^{\mathrm{T}}$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^{\mathrm{T}}X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{T}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: $\mathrm{NRHS} \geq 0$.

4: AP$(*)$ – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

On entry: the Cholesky factor of A stored in packed form, as returned by F07GDF (DPPTRF).

5: B(LDB, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

6: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07GEF (DPPTRS) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

7: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A+E)x = b$, where

if UPLO = 'U', $|E| \leq c(n)\epsilon |U^{\mathrm{T}}||U|$;

if UPLO = 'L', $|E| \leq c(n)\epsilon |L||L^{\mathrm{T}}|$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$.

Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07GHF (DPPRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07GGF (DPPCON).

8 Parallelism and Performance

F07GEF (DPPTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07GEF (DPPTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $2n^2r$.

This routine may be followed by a call to F07GHF (DPPRFS) to refine the solution and return an error estimate.

The complex analogue of this routine is F07GSF (ZPPTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 8.70 & 8.30 \\ -13.35 & 2.13 \\ 1.89 & 1.61 \\ -4.14 & 5.00 \end{pmatrix}.$$

Here A is symmetric positive definite, stored in packed form, and must first be factorized by F07GDF (DPPTRF).

10.1 Program Text

```
    Program f07gefe

!     F07GEF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: dpptrf, dpptrs, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, ldb, n, nrhs
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ap(:), b(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F07GEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
      Allocate (ap(n*(n+1)/2),b(ldb,nrhs))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
```

```
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A
!     The NAG name equivalent of dpptrf is f07gdf
      Call dpptrf(uplo,n,ap,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of dpptrs is f07gef
        Call dpptrs(uplo,n,nrhs,ap,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07gefe
```

10.2 Program Data

```
F07GEF Example Program Data
  4  2                         :Values of N and NRHS
  'L'                          :Value of UPLO
  4.16
 -3.12   5.03
  0.56  -0.83   0.76
 -0.10   1.18   0.34   1.18   :End of matrix A
  8.70   8.30
-13.35   2.13
  1.89   1.61
 -4.14   5.00                 :End of matrix B
```

10.3 Program Results

```
 F07GEF Example Program Results

 Solution(s)
                1          2
 1       1.0000     4.0000
 2      -1.0000     3.0000
 3       2.0000     2.0000
 4      -3.0000     1.0000
```